Problem 2.1

Problem 2.1. Suppose you flip four fair coins.

(a) Make a list of all the possible outcomes, as in Table 2.1.

(b) Make a list of all the different "macrostates" and their probabilities.

(c) Compute the multiplicity of each macrostate using the combinatorial formula 2.6, and check that these results agree with what you got by brute-force counting.

Consider flipping 4 fair coins. The Mathematica function Tuples[] will generate all permutations of the set {"H", "T"}.

In[1]:=
```
microstates = Tuples[{"H", "T"}, 4];
```

In[2]:=
```
TableForm[microstates]
```

Out[2]//TableForm=

H	H	H	H
H	H	H	T
H	H	T	H
H	H	T	T
H	T	H	H
H	T	H	T
H	T	T	H
H	T	T	T
T	H	H	H
T	H	H	T
T	H	T	H
T	H	T	T
T	T	H	H
T	T	H	T
T	T	T	H
T	T	T	T

In[3]:=
```
nstates = Length[microstates] (* Note this is the same as 2^4 *)
```

Out[3]=
```
16
```

Count microstates with the 'Select' function. For example, to find all states with 3 "H", we simply Select[ ] all microstates that have 3 "H"s. Then the Length[ ] function can be used to count the length of that list, i.e. 4 microstates.

In[9]:=
```
Select[microstates, Count[#, "H"] == 3 &]
```

Out[9]=
```
{{H, H, H, T}, {H, H, T, H}, {H, T, H, H}, {T, H, H, H}}
```

In[10]:=
```
Length[Select[microstates, Count[#, "H"] == 3 &]]
```

Out[10]=
```
4
```

In[11]:=
```
macrostates =
 Table[{n, multiplicity = Length[Select[microstates, Count[#, "H"] == n &]],
   multiplicity / nstates}, {n, 0, 4}]
```

Out[11]=
$$\left\{\left\{0, 1, \frac{1}{16}\right\}, \left\{1, 4, \frac{1}{4}\right\}, \left\{2, 6, \frac{3}{8}\right\}, \left\{3, 4, \frac{1}{4}\right\}, \left\{4, 1, \frac{1}{16}\right\}\right\}$$

In[12]:=
```
TableForm[macrostates,
 TableHeadings → {None, {"Number of Heads", "Multiplicity", "Probability"}}]
```

Out[12]//TableForm=

Number of Heads	Multiplicity	Probability
0	1	$\frac{1}{16}$
1	4	$\frac{1}{4}$
2	6	$\frac{3}{8}$
3	4	$\frac{1}{4}$
4	1	$\frac{1}{16}$

This is the same result we get from the combinatoric function (Binomial[n, m] in Mathematica).

In[13]:=
```
TableForm[
 Table[{m, multiplicity = Binomial[4, m], multiplicity / 2^4}, {m, 0, 4}],
 TableHeadings → {None, {"Number of Heads", "Multiplicity", "Probability"}}]
```

Out[13]//TableForm=

Number of Heads	Multiplicity	Probability
0	1	$\frac{1}{16}$
1	4	$\frac{1}{4}$
2	6	$\frac{3}{8}$
3	4	$\frac{1}{4}$
4	1	$\frac{1}{16}$